
専修大学大学院講義
GAMS文法概要(
１．GAMSはプログラミング言語である

GAMSは、プログラミング言語であり、それを使うためには、プログラムをその言語で書かかなければならない。

GAMSのプログラムは、ふつうテキスト・エディタを使って作成され、ファイルに収められる。このプログラムの入ったファイル（インプット・ファイル、拡張子“.gms”）にGAMSを走らせる。すると、アウトプット・ファイル（拡張子“.lst”）に結果が出てくる。たとえば、

GAMS modela.gms

とすると、modela.lstという名前のファイルが返されてくる。

GAMSのプログラムは、ひとつないしそれ以上の文（statements）からなる。（下の通り。）文があらわすものは、さまざまであるが、それは、データの構造を定義したり、初期値を与えたり、データの変更をしたり、シンボル間の関係（つまり方程式）を定義したりする。

各々の文の末尾には、セミコロン（；）をつける。最後の文には、セミコロンはあってもなくてもよい。（セミコロンをつねにつけることを習慣にしたほうがよいだろう。）

GAMSのインプットは、自由形式（フリー・フォーマット）である。一行の中のどこに文があってもよい。一行に複数の文があってもよいし、数行（何行でもよい）にまたがる文があってもよい。

空白とエンドオブライン（リターン）は、シンボルや単語の間に自由に使用することができる。したがって、空白行を挿入してプログラムを読みやすくすることができる。また、GAMSは、ケース・センシティブでないから、大文字、小文字の区別はない。一行は、120桁まで使える。

すべての行がGAMS言語の一部分というわけではない。行の一桁目にアステリスク（*）かダラー・シンボル（$）があれば、その行は、非言語行である。

一桁目にアステリスクがあると、その行は、コメント行であり、加工されず、コメント行として取り扱われる。

一桁目にダラー・シンボルが置かれている場合、その行には、コンパイラ・オプションが含まれているという意味となる。GAMS OUTPUTにおいては、エコー・プリントされない。

たとえば、付録の輸送問題のインプット・ファイルの冒頭には次の文が見える。

$TITLE A TRANSPORTATION MODEL

このタイトル文では、OUTPUTの各ページの先頭に、A TRANSPORTATION MODELというテキストがプリントされる。

また、次の行には、
$OFFUPPER

とあるが、この＄OFFUPPER文は、エコー・プリントが大文字、小文字をともに含むことができるようにするために必要である。

また、ブロック・デリミター（$ONTEXT……$OFFTEXT）は、コメント行を作るもうひとつの方法である。段落単位のブロック・コメントをGAMSプログラムに挿入することができる。もちろん、ダラー・シンボルは、一桁目に置く。

２．GAMSの文の分類と構成

このような非言語行を例外としてGAMSプログラムは、GAMSの文法に則ったGAMS言語で書かれることになる。

GAMSの文は、次の２つのグループのいずれかに分類される。

1． 宣言文（declaration statements）・定義文（definition statements）

2． 実行文（execution statements）

宣言文は、シンボルの集合を定める。宣言で初期値を与えてしまう場合も多い。その場合、定義文と呼ばれる。宣言・定義文には、以下のようなものがある。

ACRONYM（HERCULESでしか使われない）

SET

ALIAS

PARAMETER

SCALAR

TABLE

VARIABLE

EQUATION DECLARATION

EQUATION DEFINITION

MODEL

実行文は、データを変換するとか、モデルを解くとか、レポートを作成するとかというアクションを実行する。以下のようなものがある。

OPTION

DISPLAY

ASSINGNMENT

LOOP

ABORT

SOLVE

宣言文と定義文との区別であるが、

SET C CROPS;

は、宣言文であり、

SET C /WHEAT, CLOVER, BEANS/;

は、定義文である。前者は、集合の名前（「識別子」）をさしあたって宣言しているだけである（Cという名前は集合の名前であることを述べている）が、後者は、宣言と同時に、さらに、集合の要素（＝「ラベル」）を指定している。同様に、

PARAMETER  YIELD  Crop yield;

は、宣言文であり、YIELDという名前（識別子）をもったパラメータの存在を宣言している。後でプログラムの中にあらわれるであろう、次のような文は、定義文である。ここでは、要素と値の対が初期値として与えられている。

PARAMETER  YIELD  / WHEAT 1.5

                         CLOVER 6.5

                         BEANS 1.5 /;

GAMSプログラムにどのように文を並べるかに関しては、大きな自由度がある。ごく常識的な制約は、識別子等のシンボルは、それが使われる前にその存在が宣言されていなければならない、というものである。さらに細かく注釈すると、方程式中で使われるSET、PARAMETER等は、方程式EQUATIONが定式化（宣言と定義）される前に宣言されなければならない。しかしながら、定義は、その後でもかまわない。ただし、その方程式がSOLVE文で使われる前には定義されていなければならない。

４．リーガル・キャラクター

　GAMS言語で使用可能なキャラクタ（「リーガル・キャラクタ」）は、以下の通りである。ブレース（{…}）などが含まれていないことに注意する。

	A to Z（アルファベット大文字）
	a to z（アルファベット小文字）
	0 to 9（数字）



	&（アンバーサンド）


	“（ダブル・クオート）
	#（シャープ）



	*（アステリスク）


	=（等号）
	?（クエスチョン・マーク）

	@（アット・マーク）


	>（不等号、大なり）


	;（セミコロン）

	\（バック・スラッシュ）

	<（不等号、小なり）


	‘（シングル・クオート）

	:（コロン）
	-（マイナス）
	/（スラッシュ）

	,（コンマ）
	()（括弧）
	 （スペース）



	$（ダラー）


	%（パーセント）
	_（アンダースコア）

	.（ドット）

	+（プラス）
	


５．予約語と予約記号

GAMSは、以下の「予約語」Reserved Wordあるいは「キーワード」および予約されたシンボルをもっている。

	ABORT
	EQUATIONS
	OPTIONS
	SOLVE

	ACRONYM
	FREE
	OR
	SOS1

	ACRONYMS
	GE
	ORD
	SOS2

	ALIAS
	GT
	PARAMETER
	SUM

	ALL
	INF
	PARAMETERS
	SYSTEM

	AND
	INTEGER
	POSITIVE
	TABLE

	ASSIGN
	LE
	PROD
	USING

	BINARY
	LOOP
	SCALAR
	VARIABLE

	CARD
	LT
	SCALARS
	VBARIABLES

	DISPLAY
	MAXIMIZING
	SET
	XOR

	EPS
	MINIMIZING
	SETS
	YES

	EQ
	MODEL
	SMAX
	

	EQUATION
	MODELS
	SMIN
	


予約された非アルファベット・シンボル

	=L=
	=G=
	=E=
	=N=


６．識別子・ラベル

　識別子は、集合、パラメータ、変数等の名前である。集合の要素は、ラベルと呼ばれる。識別子とラベルについて、以下の表に示されるような制約がある。

	     
	   識別子
	 引用符なしのラベル
	 引用符つきのラベル

	  キャラクタ数
	    10
	       10
	    10

	 最初のキャラクタ
	  文字（アルファベット）
	   文字・数字
	   任意

	文字・数字以外で使用可能なキャラクタ
	   なし
	   +および-
	使用された引用符以外のすべて

	例
	○　A　A15

×　15
	· 1952-53

×　MILK&MEAT
	"YES"

'NO' '12" / FOOT'


                            　　　　　　
                      　　　               
７．テキスト
　GAMS言語中にドキュメンテーション・テキストを比較的自由に置くことができる。一行（120桁）におさまる範囲で、80桁まで、引用符（シングル・クオートあるいはダブル・クオート）をつけて、あるいは、引用符をつけずに、テキストを書き込んでおくことができる。
　テキストは、予約語と“..”と“=”で始まらず、セミコロン“;”あるいはコンマ“,”あるいはスラッシュ“/”で終わる。
　テキストは、コメント行とは異なり、たとえば、計算結果のドキュメンテーション中にそのままあらわれ、それを見やすくすることができる。
８．デリミタ（セパレータ）
　GAMS言語のデリミタ（セパレータ）にはいくつかのレベルがある。既に見たとおり、個々の文を他の文から区切るのは、セミコロン（“;”）である。しかし、次の行が予約語で始まる場合、セミコロンを使わなくてもよい（relaxed punctuation）。
　スラッシュ（“/”）は、集合の要素など、データ・リストをGAMS言語他の部分から切り離すために使われる。
データ・リスト中の区切りとしては、コンマ（“,”）、行末（end of line）が使われる。
もちろん、ブランクは、もっとも基本的なセパレータである。ブランクを0の意味でつかうことはできない。当然のことであるが、識別子や数の中にブランクを入れることはできない。
９．GAMSにおける数
　GAMSにおいては、整数、実数の区別はないので、小数点なしの数字を書いてもすべて実数として格納されることになる。
Eは、通常通り、たとえば、1E-5=1×10-5=0.00001の意味である。
GAMS言語では、実数だけでなく、INF（無限大）、-INF（マイナスの無限大）、UNDF（定義されない）、EPS（ε）、NA（not available）が使われる。ただし、UNDFは、数の入力には使えない。とくにNAは、表中にデータの欠損がある場合のデータ・エントリーに便利である。
１０．データ・タイプと定義
　GAMSには、次の6つの基本的なデータ・タイプがある。シンボル（識別子）は、そのうちのどれかに属すことが宣言されなければならない。それは、
SETS
VARIABLES
PARAMETERS（SCALAR、TABLE）
EQUATIONS
ACRONYMS
MODELS
である。
定義文は、次のような共通の構造をしている。
	PARAMETER
	A
	(I, J)
	INPUT-OUTPUT MATRIX

	データ・タイプのキーワード
	識別子
	ドメイン・リスト
	テキスト


ドメイン・リストとテキストは、常にオプションである。以下、順に説明する。
SET・SETS
S={a, b, c}と書く代わりに、
SET　S  / a, b, c /
と書く。a, b, cは、集合の要素である。順序は、関係ない。
　末尾のダンティーグの輸送問題では、

SETS
         I   Canning plants   / SEATTLE, SAN-DIEGO /

         J   Markets          / NEW-YORK, CHICAGO, TOPEKA / ;
と工場、市場の添え字の集合が定義されている。

順序に意味を持たせたいときには、特別な操作をする。すなわち、最初のSET宣言文があらわれる前に、プログラムの第1桁から始まる、コンピューターへの指示（$ONUELLIST）を行なう。
集合の要素は、キャラクター・ストリングとして格納されるので、
SET  T  / 1991, 1992, 1993, 1994, 1995 /;
のように、書いても数字ではない。なお、この場合、
SET  T  /　1991 * 1995　/;
のように、アステリスク（“*”）によって系列を表現することができる。このようなとき、GAMSは、2つのラベルの差（1桁ないし数桁、1ヶ所に限る）を見てラベル・リストを作る。もちろん、左の数字が右の数字より小さくなければならないが、たとえば、A1BC * A20BCであれば、20個のラベルが作られることになる。
　同一の集合に複数の名前を付ける必要がある場合がある。たとえば、SAMにおいて産業名を行のラベルにも列のラベルにも使いたいことがあるだろう。その場合、使われるのがALIASステートメントである。
SET  C Commodities / FOOD, CLOTHING/ ;
ALIAS (C, CP);
CPは、Cプライムのつもりである。”ALIAS”は、もともと「またの名は」という意味である。ALIASステートメントでは、集合のグループが宣言されている。
　SET文にドメインがつくのは、部分集合をあわわす場合である。

SET 

I All sectors

      / LIGHT-IND, FOOD+AGR, HEAVY-IND, SERVICES /;

T(I) Traded sectors

      / LIGHT-IND, FOOD+AGR, HEAVY-IND /;

このようにプログラムを書いておけば、GAMSは、ドメイン・チェッキングをやってくれる。

　多次元集合を記述することができる。

たとえば、以下のとおりである。

SET C Countries

  　 / JAMAICA

   　 HAITI

   　 GUYANA

    　BRAZYL /;

SET P Port

    / KINGSTON

     S-DOMINGO

     GEORGETOWN

     BELEM /;

SET PTOC( P, C) Port to country relationship

     / KINGSTON .JAMAICA

      S-DOMINGO .HAITI

      GEORGETOWN.GUYANA

      BELEN.BRAZIL /;

ラベルとラベルの間にドット（ブランクは自由）を置いて国・港のペア（tuple）を指定してゆく。ドメイン・リストから、最初のラベル、次のラベルが含まれる集合がわかる（GAMSは、ドメイン・チェッキングを行なってくれる）。さらに、次のように、括弧を使うこともできる。
SET　CR Crude oils
        / WESTTEX  West Texas
         MIDCON  Mid Continent /;
SET CI Intermediate Products
/ NAPHTHA
 GASOIL
 DISTILLATE /;
SET CICR(CI, CR) Tagges Intermediate Products
/ NAPHTHA. (WESTEX, MIDCONT)
 GASOIL. (WESTTEX, MIDCONT)
 DISTILLATE. (WESTEX, MIDCONT)/;
のように、あるいはより簡潔に
SET CICR(CI, CR) Tagges Intermediate Products
/ (NAPHTHA, GASOIL, DISTILLATE). (WESTEX, MIDCONT)/;
と書けば、
        /  NAPHTHA. WESTTEX
　　　　　NAPHTHA. MIDCONT
          GASOIL. WESTTEX
GASOIL. MIDCONT
DISTILLATE. WESTTEX
DISTILLATE. MIDCONT /;
と書いたことになる。一般にn-tuplesでも同様である。系列の場合には、括弧を使わないで、たとえば、1*3.A 1*A6. 1995*2002で同等である。

PARAMETER・PARAMETERS・SCALAR・SCALARS・TABLE
　おそらく最もGAMSらしいデータ・エントリーの方法は、TABLEステートメントによるものであろうが、GAMSのデータ・エントリーの方法には、3つある。

第1の方法は、データ・リストによる方法である。以下では、SETステートメントで添え字の集合がまず定義され、次にPARAMETERSステートメントでパラメータの値が与えられる。PARAMETERSステートメントでは、データ・リストの両側をスラッシュでくくることと要素・値の対ごとにコンマを付すか改行することを要求されるほかは、かなり自由にデータ・エントリーができる。たとえば、要素と値の間を等号で結ぶなどしてもよい。なお、すべてのパラメータに対するディフォルト値は0である。
SETS
         I   Canning plants   / SEATTLE, SAN-DIEGO /

         J   Markets          / NEW-YORK, CHICAGO, TOPEKA / ;

       PARAMETERS

           A(I)  Capacity of plant i in cases

             /    SEATTLE     350

                  SAN-DIEGO   600  /

           B(J)  Demand at market j in cases

             /    NEW-YORK    325

                  CHICAGO     300

                  TOPEKA      275  / ;
括弧およびアステリスクは、SETステートメントの場合と同様に使用できる。さらに、次のような使い方もできる。

　SET ROW   / ROW-1* ROW-10/

       COL   / COL-1* COL-10/

       SRO(ROW)  / ROW-7 * ROW-10/ ;

  PARAMETER A(ROW,COL),  B(ROW,COL),

       R(ROW) /ROW-1 22 , ROW-4*ROW-7 4.7 ,  (ROW-2, ROW-9)  2.4 )/,

       C(COL) / COL-4 17.4, COL-1 8.8, COL-9*COL-10 11.1/;

集合と結びつけられていない、言い換えればドメインなしのパラメータはSCALARステートメントによって与えることができる。

　　　　　SCALAR  F  Freight in dollars per case per thousand miles  /90/ ;  
PARAMETERステートメント（あるいはSCALARステートメント）を宣言文でとどめておき、あとで、直接割り当てる（DIRECT ASSSIGNMENT）こともできる。これがGAMSのデータ・エントリーの第2の方法である。FORTARANの代入文のようなものである。

　　　SCALAR　X ;

      X=1.2 ;

      X=X+2 ;

      SET D /NORTH, SOUTHY/ ;

      PARAMETER DA(D) /NORTH 21.2, SOUTH 3.07 / 

                    DJ(D) ;

      DJ(D) = 2.75*DA(D) ;

最後の行のassignmentの場合、添え字（index）DのついたDJ全体に値が割り当てられる。Indexed Assignmentという。Dがコントロール・インデックス（Controlling index）あるいはコントロール集合としての役割を果たしている。添え字のついた要素全体に値を割り当てるのではなく、特定の要素への割り当てには、引用符（シングル・クオートまたはダブル・クオート）を使う。

　　　A ( ‘ROW-7’, ‘COL-4’ )=-2.36 ;

コントロール・インデックス１つでは、十分な割り当てがなされない場合がある（以下を見よ）。

      B( ROW, ROW) =7.7-R(ROW);

このステートメントでは、10個の「対角要素」に値が割り当てられる。非対角要素は0のままである。そのような場合、ALIASを使う。

　　　ALIAS (ROW, ROWP);

      B( ROW, ROWP) =7.7-R(ROW);
TABLE
表形式でデータ・エントリーを行なう。以下の2次元の表（行列）の例では、パラメータDがドメインつきで宣言され、その値が表形式で与えられる。表中にブランクがあれば0と解釈される。単数形であることに注意してほしい。SET、PARAMETER、SCALARとは異なり、TABLEステートメントではひとつの識別子しか定義（宣言・初期値の割り当て）できない。

TABLE D(I,J)  Distance in thousands of miles

                        　NEW-YORK       CHICAGO      TOPEKA

          SEATTLE      　    2.5           1.7          1.8

          SAN-DIEGO     　   2.5           1.8          1.4  ;

1行ですべての列が表示できない場合、“+”記号で継続できる（以下）。

TABLE D(I,J)  Distance in thousands of miles

                        　NEW-YORK       CHICAGO      

          SEATTLE      　    2.5           1.7          
          SAN-DIEGO     　   2.5           1.8            ;

　　　　　　　　+　　　　　　　　　TOPEKA
SEATTLE 　　　　　　　1.8
SAN-DIEGO 　　　　　　1.4
VARIABLE・VARIABLES
GAMSの変数は、経済学でいう内生変数、LPにおける列・アクティヴィティー、ORにおける意思決定変数（decision variable）である。すなわち、モデルが解かれるまでは一般にその値は未知であるような変数のことである。輸送問題では、次のVARIABLEステートメントがあらわれている。Zは、スカラー量なのでドメインなしで宣言されている。GAMSの最適化モデルには、少なくともそのような変数がひとつはあらわれ、最大化ないし最小化されるべき量となる。
VARIABLES
   X(I,J)  Shipment quantities in cases
   Z       Total transportation costs in thousands of dollars ;
宣言された変数は、その型を割り当てなければならない。
変数の型　　　　　　　　　変数のとりうる範囲
FREE（ディフォールト）　　　　－∞～＋∞
POSITIVE　　　　　　　　　　　0～＋∞
NEGATIVE　　　　　　　　　－∞～0
BINARY　　　　　　　　　　　　0または1
INTEGER　　　　　　　　　　　0,1,2,...,100
最適化されるべき変数は、スカラーであり、なおかつFREE TYPEでなければならない。輸送問題のX(I, J)は、非負だから、次のようなステートメントを書く。ドメインをつけてはならない。
POSITIVE VARIABLE X ;
GAMSのパラメータとのちがいを明確にしておかなければならない。パラメータにおいては、ひとつのラベル（の組み合わせ）に対してひとつの値が対応する。しかし、変数では、4つの値が対応する。

.LO　　　下限

.UP　　　上限

.L　　　アクティヴィティのレベル

.M　　　限界値（マージナル）　あるいは双対値、費用削減値。
GAMSは、それぞれの変数と方程式とにレコードが維持されている、小さなデータベース・システムを持ったものとして設計されている。各々のレコードには上記のような4つのフィールドがある。ユーザーは、このデータベースにリード・ライトのアクセスができる。たとえば、変数の上限・下限は、変数の型に応じて自動的に設定されるが、ユーザーはそれらをオーバーライトできる。レベル変数と限界値のフィールドに関しては、ユーザーが初期値を与えることができるが、そのあとは、ソルバーがコントロールする。非線形計画について、レベル変数の初期値を設定できる意味は大きい。

EQUATION・EQUATIONS

Equation文では、宣言と定義とを別の文で行わなければならない。宣言のフォーマットは、GAMSの他の識別子と同様である。輸送問題には、3つのEQUATION DECLARATIONがある。
  EQUATIONS
       COST        Define objective function
       SUPPLY(I)   Observe supply limit at plant i
       DEMAND(J)   Satisfy demand at market j ;
GAMSにおけるEQUATIONは、不等式の意味をも含んでいることに注意。また、COSTは、ドメインがないので単一方程式である。
方程式の定義EQUATION DEFINITIONは、シンボル間の関係が定義される。以下の構成要素からなる。
1． 定義される方程式の名前。

2． ドメイン。
3． （オプショナル）ドメイン制限条件。ダラー・コントロールで行なう。
4． シンボル“..”。
5． 左辺の式。

6． 関係演算子。

=L=（すなわち“≦”）

=E=（すなわち、“＝”）

=G=（すなわち、“≧”）。

7． 右辺の式。

輸送問題には、以下のステートメントがあらわれている。
COST  ..        Z  =E=  SUM((I,J), C(I,J)*X(I,J)) ;

　  SUPPLY(I)  ..   SUM(J, X(I,J))  =L=  A(I) ;

    DEMAND(J)  ..   SUM(I, X(I,J))  =G=  B(J) ;
Zは、輸送問題の目標関数である。目標関数は、EQUATION DEFINITIONのうちのどれかにあらわれなければならない。

GAMSの方程式では、右辺と左辺に同じ変数があらわれてもかまわない。GAMSの側で変換してくれる。EQUATION DEFINITIONは、含まれる変数やパラメータが既に宣言されていれば、GAMSインプットのどこに置いてもかまわない。

MODEL
　MODELステートメントは、方程式をグループ分けし、それに名前をつけ、SOLVEのプロセスに進むことができるようにする。モデル文は、以下の構成をもつ。テキストは、オプション、モデル名は、GAMSの識別子
MODEL  　　TRANSPORT   　　A transportation model 　　  / ALL / ;
キーワード　　　モデル名　　　　　　テキスト　　　　　　　　/モデル・リスト/
１１．GAMS言語における数式と標準関数
　実行文の説明に入る前に、GAMSにおける演算記号と標準関数を導入する。GAMS言語では以下の演算記号を使用する。演算の優先順位も、以下の順である。括弧がないかぎり、表中の上の演算が優先して実行される。
	  　**
	   べき乗（実数べき）


	   *    / 
	  乗算・除算

	  +　　-
	  加算・減算


GAMS言語には、次のようなIndexed Operationがある。
SUM
PROD
SMIN
SMAX
GAMSの和記号（サメーション）は、ふたつのアーギュメントをもった演算であり、
SUM( index of summation, summand)
の形式で使う。たとえば、SUM(J, X (I, J ) )は、　

　のことである。ネステド・サメーションも、たとえば、SUM( I, SUM(J, C (I, J )* X ( I, J ) ) )のように書ける。

　積記号は、SUMのかわりにPRODを使う。たとえば、PROD(J, X (I, J ) )は、

のことである。

　SMIN、SMAXの演算子は、たとえば、SMIN( I, X( I, J ))のかたちで
[image: image4.wmf]ij

の最大、最小を与える。
GAMSの標準関数

	名前
	説明
	説明
	argの数

	ABS
	Absolute value
	| arg |
	1

	ARCTAN
	Arctangent
	arctan(arg); 結果は、ラジアン。
	1

	CEIL
	Ceiling
	最小の整数≧arg
	1

	COS
	Cosine
	cos(arg); argは、ラジアン。
	1


	ERRORF
	Error function
	－∞からargまでの標準正規分布の積分値。
	1

	EXP
	Exponential
	
[image: image5.wmf]arg

	1

	FLOOR
	Floor
	最大の整数≦arg
	1

	LOG
	Natural logarithm
	argの底
[image: image6.wmf]e

の対数値。
	1

	LOGIO
	Common logarithm
	argの底10の対数値。
	1

	MAPVAL
	Mapping function
	特別な値に一意数を割り当てる。
	1


	MAX
	Largest value
	max(arg1, arg2, arg3,…)
	>1

	MIN
	Smallest value
	min(arg1, arg2, arg3,…)
	>1

	MOD
	Remainder
	arg-trunc(arg1/arg2)×arg2
	2

	NORMAL
	Normal random
	平均がarg1、標準偏差がarg2である（疑似）正規乱数。方程式にあらわれてはならない。”seed”は、OPTION SEED=numberによってリセットできる。シードは、ひとつづきのコンパイレーションで維持されている。
	2

	POWER
	Integer power
	
[image: image7.wmf]2

arg

1

arg

。ここで、arg2は、整数。
	2

	ROUND
	Rounding
	round(arg)は、最も近い整数に丸められもの。

round(arg1, arg2)は、小数点以下arg2桁にarg1を丸めたもの。
	1ないし2

	SIGN
	Sign
	　　　　　+1　if　arg>0

sign(arg)= -1  if  arg<0

          0   if  arg=0
	1

	SIN
	Sine
	Sin(arg);argは、ラジアン。
	1

	SQR
	Square
	arg×arg
	1

	SQRT
	Square root

	1

	TRUNC
	Truncation
	sign(arg)×floor(abs(arg))
	1

	UNIFORM
	Uniform random
	Arg1とarg 2の間の一様乱数。方程式にあらわれてはならない。”seed”は、OPTION SEED=numberによってリセットできる。シードは、ひとつづきのコンパイレーションで維持されている。
	2


１２．実行文
SOLVE
　GAMSによって解くことのできる問題はさまざまある。しかし、解くためには、問題の型を知らなければならない。また、解き方は一通りではないから、0-1変数や整数変数が含まれていても、MIP（Mixed integer Programming）問題として解くか、RMIP（Relaxed Mixed integer Programming）問題として解くかは、選択の余地がある。SOLVEステートメントに必要となる、問題類型とその識別子は、以下の通り。ただし、GAMSそのものが問題を解くのではなく、正しく定式化された問題をソルバー・プログラムに送る。
LP　線形計画。
NLP　非線形計画。　モデルに「滑らかな」関数しかあらわれない非線形計画。
DNLP　非連続微分非線形計画。「滑らか」でない関数があらわれる非線形計画。
RMIP　緩和混合整数計画。非連続変数を含むが非連続制約が緩和されている。整数、0-1変数は上下限内の任意の値を取ることができる。
MIP　混合整数計画。
RMIDNLP　非連続微分緩和混合整数計画。
MIDNLP　非連続微分混合整数非線形計画。
SOLVEステートメントのシンタックスは単純である（以下の通り）。
SOLVE     TRANSPORT    USING    　LP      MINIMIZING      COST;
キーワード 　モデル名　　キーワード　モデル・タイプ　最大・最小の指定　目標変数　　　　
ディレクション（最大・最小の指定）は、“MAXIMIZING”、“MINIMIZING”というもうひとつのキーワードを使う。
   目標変数（Objective Variables）は、既に述べた通り、スカラーでFREE型でなければならない。もちろん、モデル中の少なくともひとつの方程式にあらわれなければならない。
　なお、ソルバー・プログラムを自由にGAMSにリンクすることができる。そのためのドキュメントが用意されているし、そのようなリンクのための既存のソースコードは、公開されている。
ソルバーの一覧表
モデル・ソルバーの型
         ソルバー・サブシステム

   LP
 BDMLP、MINOS 5、ZOOM、MPSX、SCICONIC、APEX IV

  NLP
 MINOS 5、CONOPT、GRG 2、NLPNETG、NPSOL

  MIP
  ZOOM、MPSX、SCICONIC、APEX IV

DISPLAY
ソルバーのアウトプットを表示させるには。DISPLAY文を使う。輸送問題には、次のステートメントがあらわれている。変数を表示させるには、どの値か（.UP、.LO、.L、.M）をしめさなければならない。

DISPLAY X.L, X.M ;
GAMSアウトプットには、以下のように計算結果が示される。

VARIABLE  X.L           SHIPMENT QUANTITIES IN CASES

             NEW-YORK     CHICAGO      TOPEKA

SEATTLE        50.000     　300.000

SAN-DIEGO     275.000                 　　275.000

VARIABLE  X.M           SHIPMENT QUANTITIES IN CASES

              CHICAGO      TOPEKA

SEATTLE                     0.036

SAN-DIEGO       0.009
１３．GAMSアウトプット

 エコー・プリント、（エラーメッセージ）、レアフェンス・マップ、方程式リスト、モデル統計、ステータス・レポート、ソリューション・レポートなどが出力される。

参考文献

A.Brooke, D.Kendrick, and A.Meeraus, GAMS : A User's Guide, Release 2.25 ,Scientific Press.

付録　1　ダンティーグの輸送問題のGAMSインプット・ファイル

ダンティーグ『線形計画とその周辺』（George B. Dantzig, Linear Programming and Extensions, Princeton University Press, 1963）は、現在でも線形計画法（LP）の基本文献のひとつである。その第3章第3節に輸送問題の数値例が与えられていた。以下のプログラムでは、その数値例が若干変更されて使用されている。以下のインプット・ファイルを読みとることは容易であり、問題そのものの説明は不要と思われる。

$TITLE  A TRANSPORTATION PROBLEM

$OFFUPPER

  SETS

       I   canning plants   / SEATTLE, SAN-DIEGO /

       J   markets          / NEW-YORK, CHICAGO, TOPEKA / ;

  PARAMETERS

       A(I)  capacity of plant i in cases

         /    SEATTLE     350

              SAN-DIEGO   600  /

       B(J)  demand at market j in cases

         /    NEW-YORK    325

              CHICAGO     300

              TOPEKA      275  / ;

  TABLE D(I,J)  distance in thousands of miles

                    NEW-YORK       CHICAGO      TOPEKA

      SEATTLE          2.5           1.7          1.8

      SAN-DIEGO        2.5           1.8          1.4  ;

  SCALAR F  freight in dollars per case per thousand miles  /90/ ;

  PARAMETER C(I,J)  transport cost in thousands of dollars per case ;

            C(I,J) = F * D(I,J) / 1000 ;

  VARIABLES

       X(I,J)  shipment quantities in cases

       Z       total transportation costs in thousands of dollars ;

  POSITIVE VARIABLE X ;

  EQUATIONS

       COST        define objective function

       SUPPLY(I)   observe supply limit at plant i

       DEMAND(J)   satisfy demand at market j ;

  COST ..        Z  =E=  SUM((I,J), C(I,J)*X(I,J)) ;

  SUPPLY(I) ..   SUM(J, X(I,J))  =L=  A(I) ;

  DEMAND(J) ..   SUM(I, X(I,J))  =G=  B(J) ;

  MODEL TRANSPORT /ALL/ ;

  SOLVE TRANSPORT USING LP MINIMIZING Z ;

  DISPLAY X.L, X.M ;
付録　2　輸送問題についてのノート

線形計画法の特殊ケースである輸送問題は、もちろん、よく知られた線形計画問題の特殊ケースである。

輸送問題は、カントロヴィッチ（L.V. Kantorovich）、ヒッチコック（Frank L. Hitchcock）、クープマンス（T.C. Koopmans）によって第二次世界大戦中およびその直後に定式化された。

輸送問題は、シンプレックス法（単体法）がダンティーグによって確立される以前に、本質的には、シンプレックス法と同等の方法をたかだか紙と鉛筆によって解くことができることが知られていた。いわゆるDOSSO、すなわち、『線形計画と経済分析』（Robert Dorfman, Paul Samuelson & Robert Solow: Linear Programming and Economic Analysis,McGraw-Hill,1958）では、第５章を輸送問題に充て、この略式解法を紹介していた。

　以下では、DOSSOにおける簡単な例示とともに輸送問題を定式化してみよう。まず、次のデータは所与である。iを工場（たとえば、缶詰工場）の添字、jを市場の添字とする。

意志決定の対象となる変数は、

である。この変数には、非負制約がある。すなわち、

工場iの供給制約、市場jの需要制約がある。

目的関数は、輸送費用の最小化である。（単位は、＄K）
…………(3)
DOSSOの数値例は、工場が３ヶ所（A,B,C）にあり、市場が5ヶ所（1,2,3,4,5）にあるケースである。生産能力、所要量（単位トン）、１トン当たりの輸送費用（単位ドル）が次の表のように与えられていることにする。

	販売地区
	
	工場
	
	所要量

	　　　　
	A
	B
	C
	

	1
	$10
	$20
	$30
	25

	2
	$15
	$40
	$35
	115

	3
	$20
	$15
	$40  
	60

	4
	$20　
	$30
	$55
	30

	5
	$40   
	$30  
	$25
	70   

	生産能力
	50
	100
	150
	300


スターティング・ルール（最初の実行可能基底解の求め方）は、複数ある。いずれも、計算をほとんど必要とせず表を見るだけで最初の実行可能基底解が得られる。ここでは、DOSSOのものを紹介する。工場Aから始めよう。Aの列で最小の輸送費用を与えるcijをもつ行（市場1 cA1=15）を見つける。そこで、工場Aの供給能力を市場1に充てる。まだ25トン供給能力が残っているから、次に輸送費の安い市場２に充てる（cA2=15）。これで工場Aは、供給能力を使い果たしたが、まだ、市場2の需要は、90残っている。そこで、市場2の列を観察し、次に輸送費の安い供給源を探す。工場Cが安いので、工場Cから90トン市場2に輸送することにする。市場2は、これで所要量を満たしたことになる。

次に工場Bについて同じ作業をする。
まず、市場3に供給し、供給余力40は、市場4、5どちらでもよいが、市場5に回す。市場5の需要がまだ30あるからそれを満たすために工場Cから30輸送する。工場Cの供給余力30は、もちろん市場4に輸送する。DOSSOの設例では、各工場の供給能力の合計と各市場の需要量の合計は等しく設定されていることに注意する。このことは、8つの制約式のうち、独立な方程式は、7つしかないことを意味する。上のプロセスでも0でないxijが7つしかないことに注意する。実は、このプロセスは、実行可能基底解を見いだすためのものである。

第１次出荷計画
	販売地区
	
	工場
	
	所要量

	　　　　
	A
	B
	C
	

	1
	25
	
	
	25

	2
	25
	
	90
	115  

	3
	
	60
	  
	60

	4
	　
	
	30
	30

	5
	   
	40  
	30
	70   

	生産能力
	50
	100
	150
	300


　このように得られた「第1次出荷計画」が基底解となっていることをどのように確かめればよいのだろうか。あるいは、第1次出荷計画を作成するための以上の作業のキー・ポイントはどこにあったのだろうか。次の2つの図を比べてみよう。図の読みとり方はほとんど自明である。上の図では、すべての「辺」と「点」がひとつながりになっており、下の図はそうでない。3つの工場、5つの市場のすべてをひとつながりにするためには、少なくとも7つの辺が必要であることにも注意する。グラフ理論によれば、8つの点、7つ（点の数－1）の辺からなるグラフが連結（ひとつながりの）グラフであることとサイクルをもたないことは同値である。サイクルをもたない連結グラフは「木」と呼ばれる。木においては、辺の数は、点の数－1である。要するに、第1次出荷計画を作成するプロセスは、「木」を作る作業そのものなのである。上の図のグラフは木であり、下の図のグラフは木でない。下の図のグラフは、木でない（連結でない）から、サイクルを持っていることにも注意する。木においては、それに含まれないひとつの辺を付け加えるとちょうどひとつのサイクルをもつようになることも知られている。

たとえば、（B-1）という計画に含まれない工場－市場の（任意の）組み合わせを新しい「辺」として書き加えてみると、(B-1)（1-A）（A-2）（2-C）（C-5）（5-B）という第1次出荷計画に含まれている「辺」をたどってBに戻ってくることができる。見方を変えていえば、Bから1へ向かう2通りの径路がある。直接径路と間接径路である。
　直接径路：（B-1）
　間接径路：（B-5）（5-C）（C-2）（2-A）（A-1）
このように、どの工場－都市のペアに対しても、木に含まれる径路をたどることによって両者をつなぐことができる。これは、この木がベクトル空間の基底であることをも意味する。

　式によってこのことを見てみよう。輸送問題を線形計画問題として定式化しなおしたのが第(4)式である。工場の生産能力制約をあらわす不等式の向きを市場需要制約をあらわす不等式の向きとあわせるために符号を工夫している。これは、線形計画における典型的な工夫である。また、係数行列を観察すると、縦にすべての式を足し合わせるとすべて0になっていることがわかる。つまり、すべての式が独立であるわけではない。そこで、最後の式を省略することにする。（この数値例の場合、厳密な不等式はありえないことは最初から明らかである。）
　木を作るように選ばれた辺をベクトルと解釈することができる。ベクトルの最初の3つの要素は、工場、あとの4つの要素（8番目の、最後の要素は省略）は都市をあらわす。そこから出発する工場には、マイナス1、そこに到達する都市には、プラス1の数字を記入し、その他の要素は、0を記入する。

　第1次出荷計画に含まれない、たとえば、（B-1）は、この基底ベクトル群の1次結合として表現できる。第（5）式にそのことを示した。右辺のベクトルは、その作り方から右辺の行列A０の逆行列を左辺のベクトルの左側から掛けたものである。

第（4）式を等式にして、基底部分、非基底部分に係数行列を分割し（A０、A１）、プロセスのレベルをあらわすベクトルｘも2つに部分（ｘ０、ｘ１）にわけ、さらにさきに述べたように最後の式を省略すると、右辺の定数ベクトルをｂとすることにより、次のように書ける。

A０の逆行列を左側から掛け合わせると

ここで、目標関数を

と書くことにすると、

上式でｘ１の係数は、直接費用（c１）部分と木に含まれる径路を利用した間接径路に拠った場合の間接費用との差になっていることに注目する。この係数がすべてプラスなら現在の木が最小費用をあたえていることになる。マイナスの要素があれば、現在の選択には改善の余地がある。

ところで、第（5）式の右辺のベクトルについて述べた注意により、

は、間接径路すなわち、（B-5）（5-C）（C-2）（2-A）（A-1）上で工場から都市に向かうプロセスにはその費用にプラスの符号を付し、都市から工場に向かう逆のプロセスには、マイナスの符号を付して合計したものである。たとえば、この場合、直接費用は、$20、間接費用は、+$30-$25+$35-$15+$10=$35であり、直接費用が間接費用より安い。したがって、（B-1）を導入することにより第1次出荷計画は改善することができる。第1次出荷計画に含まれない他の組み合わせについても間接費用が計算できる。下の表では、間接費用を一覧表にし、直接費用－間接費用も表にしている。

第1次出荷計画の間接費用

	
	A
	B
	C

	1
	*
	35
	30

	2
	*
	40
	*

	3
	-10
	*
	10

	4
	35
	60
	*

	5
	5
	*
	*


直接費用と間接費用の差（第1次出荷計画）
	
	A
	B
	C

	1
	*
	-15
	0

	2
	*
	0
	*

	3
	30
	*
	20

	4
	55
	-30
	*

	5
	35
	*
	*


直接費用と間接費用の差（マージナル）の表を見ると、（B-1）より（B-4）を出荷計画に組み込んだ方がよさそうに見えるので、（B-4）を導入することにしよう。間接径路は、（B-5）（5-C）（C-4）であるから、（B-4）を1単位導入するごとに、第1次出荷計画から（B-5）を1単位減らし、（C-5）を1単位増やし、（C-4）を1単位減らさなければならない。（B-5）のレベルが40、（C-4）のレベルが30であるから、このプロセスは（C-4）のレベルをちょうど0にするところまでしか進められない。そうすると、次のような第2次出荷計画が得られる。

第2次出荷計画

	販売地区
	
	工場
	
	所要量

	　　　　
	A
	B
	C
	

	1
	25
	
	
	25

	2
	25
	
	90
	115  

	3
	
	60
	  
	60

	4
	　
	30
	60
	30

	5
	   
	10  
	
	70   

	生産能力
	50
	100
	150
	300


再び間接費用の表および直接費用と間接費用の差の表を作成すると、以下の通りになる。
第2次出荷計画の間接費用

	
	A
	B
	C

	1
	*
	35
	30

	2
	*
	40
	*

	3
	-10
	*
	10

	4
	5
	*
	25

	5
	5
	*
	*


直接費用と間接費用の差（第2次出荷計画）
	
	A
	B
	C

	1
	*
	-15
	0

	2
	*
	0
	*

	3
	30
	*
	30

	4
	15
	*
	30

	5
	35
	*
	*


マージナルがマイナスのエレメントはひとつしかないから、今度は、（B-1）を導入して第3次計画を作成する。

第3次出荷計画
	販売地区
	
	工場
	
	所要量

	　　　　
	A
	B
	C
	

	1
	15
	10
	
	25

	2
	35
	
	80
	115  

	3
	
	60
	  
	60

	4
	　
	30
	70
	30

	5
	   
	  
	
	70   

	生産能力
	50
	100
	150
	300


第3次出荷計画の間接費用
	
	A
	B
	C

	1
	*
	*
	30

	2
	*
	25
	*

	3
	5
	*
	25

	4
	20
	*
	40

	5
	5
	15
	*


直接費用と間接費用の差（第3次出荷計画）
	
	A
	B
	C

	1
	*
	*
	0

	2
	*
	15
	*

	3
	15
	*
	15

	4
	0
	*
	15

	5
	35
	15
	*


 上の表から知られる通り、第3次出荷計画では、マイナスのマージナルは存在せず、これで最適な出荷計画が得られていることがわかる。（ただし、この問題の最適解は一意ではない。）

練習問題　DOSSOの輸送問題をGAMS言語で書け。

付録　3　線形計画問題における双対と限界費用

DOSSOにならい、食餌問題（Diet Problem; diet=常食）を例に線形計画問題を定式化する。5種類の食品（1、2、3、4、5）があり、その中から適当なものを選んでメニューを作り、必要なカロリーとビタミンを確保するという問題である。DOSSO第2章第3節では、以下の数値例が与えられている。
	
	食

た

分
	品

り
	1

の
	単

栄
	あ

養
	

	
	1
	2
	3
	4
	5
	標準量

	栄養素：
	
	
	
	
	
	

	カロリー
	1
	0
	1
	1
	2
	700

	ビタミン
	0
	1
	0
	1
	1
	400

	価格
	2
	20
	3
	11
	12
	


この数値例のもとで食餌問題とは、次のような費用最小化問題をさす。

線形計画問題は、この食餌問題のように、非負制約をもった変数が一次不等式であらわされる制約を満たしながら、ある目的関数を最小化あるいは最大化する問題のことである。
線型計画法に関して、双対性“duality”として知られていることがらがある。すなわち、線形計画問題に対して双対問題と呼ばれる問題が構成でき、原問題と以下のような関係をもっていることが知られている。
1． 双対問題は、原問題の制約式1本について1つづつの変数をもっている。

2． 双対問題は、原問題の変数1つについて1つづつの制約式をもっている。
3． 最大化問題の双対は、最小化問題であり、最小化問題の双対は最大化問題である。

4． 原問題の目的関数の係数は、双対問題の制約式の定数項としてあらわれ、原問題の制約式の定数項は、双対問題の目的関数の係数になっている。

5． 原問題と双対問題とでは、制約条件の係数の配列状態は、互いに転置行列の関係にある。
6． 原問題と双対問題の不等号の向きは、逆になっている。

7． 原問題でも双対問題でも、変数の非負制約は変わらない。

8． （双対定理）原問題、双対問題がともに制約条件をみたす解（実行可能解）をもてば、ともに最適解をもち、両問題の目的関数の最大値と最小値は等しくなる。

具体的には、上の食餌問題を原問題とする双対問題は、以下の通りである。

双対問題の変数の意味は何だろうか。u1 、u2をカロリーの（潜在）価格、ビタミンの（潜在）価格と考えれば、z’ を価値額とみなすことができるであろう。食品の価格は、少なくともその食品に含まれる栄養素の（潜在）価格の合計額と同じである必要があると考えられるであろう。
ここで、「限界費用」という概念を導入する。所与の実行可能解を出発点として、ビタミンの摂取量をかえずにカロリーの摂取量を1単位増やした場合（カロリーの摂取量をかえずにビタミンの摂取量を1単位増やした場合）に生じる食餌費用の増加分を限界費用と呼ぶ。実は、双対変数は、限界費用のことなのである。

たとえば、100単位の食品4および300単位の食品5からなる制約条件を満たす（実行可能な）食餌を出発点としてみよう。ちょうど、カロリー1単位だけを摂取するための食品、また、ちょうど、ビタミン1単位だけを摂取する食品の組み合わせを次式のように見いだすことができる。マイナスの要素は、たとえば、最初の式の場合、次のように説明すればよいであろう。食品5を1単位増やし、同時に食品4を1単位減らすと費用は、12ドル－11ドル＝1ドルかかる、と。同様にあとの式では、限界費用＝10ドルと計算できる。

この場合は、簡単に一意に限界費用が確定できるが、出発点となる食餌が非効率なものであったりすると、限界費用がマイナスになってしまう可能性もあるが、そのような食餌は考察からはずしてよいであろう。
上式は、各栄養素をいわば個別に購入する方法を示している。また、明らかに、

である。各栄養素の限界費用を並べてベクトルu’ と書くことにすると、

付録2の用語で間接費用と呼んでいたものは、このu’ にA の列ベクトルを左から乗じたものである。最小化問題の最適解では、直接費用－間接費用≧0となるはずだから、このu’ は、次式を満たす。

ところで、原問題（最小化問題）の制約式

と上の式を比べ、

であることがわかる。双対問題（この場合、最大化問題）の最適解における目的関数u’bの値は、原問題（最小化問題）の任意の実行可能解における目的関数c’x の値以下である。
原問題（最小化問題）の最適解では、直接費用－間接費用≧0であることはすでに見たとおりである。また、正のレベルであるプロセスでは、直接費用－間接費用＝0だから、「双対定理」が成り立つ。つまり、最小化問題の最適解における目的関数の値と最大化問題の最適解における目的関数の値とは等しい。
なお、輸送問題（付録 2）について、原問題を再掲すると、
[image: image33.wmf]å

å

j

ij

ij

i

x

c

Minimise


[image: image34.wmf]å

£

j

i

ij

i

k

x

　

　

　　

all

for

,

s.t.

[image: image35.wmf]å

³

i

j

ij

j

r

x

　

　

　　

all

for

,


双対問題は、以下のような問題である。

双対問題の制約式は、任意の工場iと販売地区jの組み合わせに対して、その地区での価値が工場での価値と輸送費の和より大きくなってはならないことを意味する。独立な式が足りないので、未知数のうちどれかひとつに任意の値（たとえば、uC）を与えればよい。結果は、次表。
	
	工
	
	場
	需
	
	要
	
	地

	
	A
	B
	C
	1
	2
	3
	4
	5

	変数
	uA
	uB
	uC
	v1
	v2
	v3
	v4
	v5

	トンあたりドル
	20
	10
	0
	30
	35
	25
	40
	25

輸送問題（第1次出荷計画）のグラフ

不適切な第1次出荷計画のグラフ

( 本稿は、専修大学大学院経済学研究科の講義原稿として用意したものである。ダンティーグによる輸送問題（付録1を見よ）をGAMS言語の説明に使っていることもあり、輸送問題について、あるいは、線形計画問題一般についてのself-containedの解説を含んでいる。


� 言うまでもないが、わが国のパソコンでは、円記号とバックスラッシュとが区別されない。


� 「ドット」は、「中黒」でなく、「ピリオド」である。


�　整数べきは、標準関数として用意されているPOWERを使う。


�　アメリカン・スペリングであることに注意。


� どうせ、ここで、工場Cが登場しているのだから、工場Cから輸送費の安い5に工場の能力いっぱいまで（150-90=60）割り当てて、残りの分を工場Bでまかなうことも合理的に思える。この方法だと、本文で説明したDOSSOの「第1次出荷計画」を改善した「第2次出荷計画」に対応する。


� たとえば、J. A. Bondy & U. S. R. Murty／立花俊一・奈良知恵・田澤新成訳『グラフ理論への入門』、共立出版、1991年、22-23頁。


� たとえば、二階堂副包『経済のための線型数学』培風館、1961年を見よ。

